

Geometrically Restrained INorganic Structure Prediction : GRINSP

Armel Le Bail,

Université du Maine, Laboratoire des Fluorures, CNRS UMR 6010, Avenue O. Messiaen, 72085 Le Mans 9, France - Email : alb@cristal.org ; WWW: <http://www.cristal.org/>

Introduction

With the idea in mind that the structure of a compound like τ -AlF₃ [1], unique example of a new 3D 6-connected network (no other isostructural MX₃ known up to now), should have been predicted before the laborious structure determination succeeded (finalized in 1992, from powder diffraction data), one can decide to write the prediction software. Which language would be more appropriate ? This may depend on the algorithm retained, but generally it mainly depends on the scientist knowledge. Would it be reasonable to propose the subject for a thesis or a post-doc ? Subject : "You will have to demonstrate that MX₃ compounds, built up from [MX₆] regular octahedra, sharing exclusively corners, can be predicted just as zeolites are predictable". It would be a bit difficult to find the good candidate. Conclusion, you will have to do it by yourself. It is your idea after all... The programming language will be the language which you best know, appropriate or not : Fortran77 in that case. A programming language may be considered as not obsolete if a compiler still exists for building the executable on recent computers. The project, imagined in 1992, was frozen up to the end of 2003, till personal computers became fast enough (the frequency of microprocessors increased by a factor 100 in the 1992-2003 range, the number of transistors increased as well a lot) for tempting to solve the problem by a Monte Carlo approach.

When the realization of such a project starts, you are not even sure to succeed. Very recent publications show that, concerning 3D 4-connected nets, systematic enumeration is now based on advances in mathematical tiling theory [2-4]. Unless you are a brilliant mathematician, this may not be of a great help when trying to transpose to 3D 6-connected nets (so, it is verified again here that Monte Carlo is the solution retained by the illiterates...). Previous works on hypothetical zeolites were made by using classical physical model building [5] during the past 60 years, or simulated annealing [6]. Many recent works in inorganic structure prediction (as well as organic and organometallic) have produced huge quantities of hypothetical compounds (using commercial packages as CERIUS, etc), there is no room here for citing them all. But there was no systematic recent work on MX₃ compounds, apart from the famous book of Wells [7] about three-dimensional nets and polyhedra. Predictions of new MX₃ compounds are non-existent, if one excludes the obvious models built up by intergrowth of known structures (perovskite, Hexagonal Tungsten Bronze type - HTB, etc).

Algorithm

It was chosen to manage the Monte Carlo generation of 3D nets by using geometrical restraints established from the interatomic distances in known materials. So, this is absolutely not an *ab initio* approach of the structure prediction problem. Multiple difficulties were solved one after the other during the writing of the source code. A primarily version of the program named GRINSP (Geometrically Restrained INorganic Structure Prediction) [8] was limited to the building of tetrahedra linked by corners (3D N-connected nets with N = 4) and to the P1 space group, because this was the more easy for testing the feasibility, all being more simple to develop in P1 on the point of view of writing the code. Then, obtaining encouraging results (150 hypothetical zeolites built in P1), the project was generalized to various values of N (3, 4, 5, 6), and even to mixtures of M and M' cations with different coordinations, in any space group [8]. In a N-connected 3D net of M atoms, each M atom is connected to N other M atoms through X atoms, giving formulations MX₂ for N = 4 (tetrahedra connected by corners as in SiO₂ polymorphs and zeolites), MX₃ for N = 6 (octahedra connected by corners) or M₂X₃ for N = 3 (triangles sharing corners as in B₂O₃), etc. The X atoms have to lie at positions close to the mid point of two M atoms. Therefore, the key of the algorithm is to concentrate on the M atoms first. For such N-connected 3D nets, if a model shows all M atoms surrounded by the expected number of M atoms (3, 4, 5 or 6), then

this model is a possible solution. This expected number of neighbours (mvexp below) is checked frequently in the GRINSP program code, each times a new set of M atoms is added either on a general or on a special position of a selected space group :

```

C We expect to have each M surrounded by 3, 4, 5 or 6 M at first
C M-M distances, then the total of expected neighbours is :
mvexp=ncop(mcop(nl(1)))*itot
C See neighbouring -
C Are some M atoms neighbouring already completed ?
call complet(itot,x,met,f,g,nv,mv,mv2tot,ncop,mcop,ibad)
C If mvexp-mv2tot = icon then store the result
iresult=mvexp-mv2tot
if(iresult.eq.icon)go to 1002
C Place the next atom of type M at acceptable distances
C considering first and second M-M neighbours

```

The Fortran77 language is quite easy to understand for people speaking english. GRINSP contains a lot of comments (lines starting by C). Either calculations or text sequence manipulations and comparisons can be done (etc), you are limited only by your imagination. The subroutine `complet` in the lines of code above is too large for showing it all here. The full GRINSP code contains more than 3000 lines, only some parts will be selected and listed in this article.

In the purpose to obtain this adequation (`iresult=icon=0` above) between the expected number of neighbours and the calculated number, the model is built sequentially, adding one M atom after the other. GRINSP does not work by applying simulated annealing to a starting random configuration. Version 1.00 works schematically as follows, by using the Monte Carlo method :

- Manual selection of the restraints on cell parameters, of restrained interatomic distances, of the type(s) of coordinations, and of the space group. Then the Monte Carlo process starts.
- Random selection of the cell parameters inside of the predefined range. The random generator subroutine in GRINSP is `randi` (see below), returning a value between 0. and 1., called very often in the program; `nsym = 1` is corresponding to the cubic case, `nsym = 2` corresponds to tetragonal, `nsym = 3` is hexagonal or trigonal, `nsym = 4` is orthorhombic, etc, other variables (`a`, `b`, `c`, `alp`, `bet`, `gam`) are self-explicit for crystallographers :

```

C Define the cell parameters
a=(amax-amin)*randi(iseed)+amin
if(nsym.eq.1)then
b=a
c=a
go to 8000
endif
c=(cmax-cmin)*randi(iseed)+cmin
if(nsym.eq.2.or.nsym.eq.3)then
b=a
go to 8000
endif
b=(bmax-bmin)*randi(iseed)+bmin
8000 continue
if(nsym.le.5)then
alp=90.
bet=90.
gam=90.
if(nsym.eq.3)gam=120.
if(nsym.eq.5)bet=betmin+betd*randi(iseed)
go to 8500
endif
alp=alpmin+alpd*randi(iseed)
bet=betmin+betd*randi(iseed)
gam=gammin+gamd*randi(iseed)
8500 continue

```

- Random positioning of a first cation M (or M') of the future M_xX_y (or $M_xM'_yX_z$) compound on a general or special position, itself selected randomly.

```

C Place the first atom of first type (M1)
4502  itot=0
      nl(1)=1
      nl(2)=0
C Decide at random for the polyhedra type
      mcop(nl(1))=int(randi(iseed)*float(npol)+1.)
      if(mcop(nl(1)).gt.npol)mcop(nl(1))=npol
C Decide for the Wyckoff position selected between np1 and npos
      mwyc(nl(1))=int(randi(iseed)*float(npos-np1)+np1)
C Decide for the atomic coordinates
      x0=randi(iseed)
      y0=randi(iseed)
      z0=randi(iseed)
      gen=gen+1.
C Extend to all positions corresponding to mwyc(nl(1))
      is=mwyc(nl(1))
      it0=itot+1
      it01=it0-1+nas(is)
      if(it01.gt.65)go to 4502
      DO 299 k=1,nas(is)
        itot=itot+1
        it1(itot)=it0
        it2(itot)=it01
        ist(itot)=is
        mcop(itot)=mcop(nl(1))
        x(itot,1,1)=x0*smt(is,k,1,1)+y0*smt(is,k,1,2)+z0*smt(is,k,1,3)
        1+tt(is,k,1)
        x(itot,1,2)=x0*smt(is,k,2,1)+y0*smt(is,k,2,2)+z0*smt(is,k,2,3)
        1+tt(is,k,2)
        x(itot,1,3)=x0*smt(is,k,3,1)+y0*smt(is,k,3,2)+z0*smt(is,k,3,3)
        1+tt(is,k,3)
299    continue
C Now we have itot atoms already, but...
C Avoid short distances
      if(itot.eq.1)go to 302
      do 300 mm1=1,itot-1
        do 300 mm2=it1(itot),itot
          if(mm1.eq.mm2)go to 300
          p1=abs(x(mm1,1,1)-x(mm2,1,1))
          q1=abs(x(mm1,1,2)-x(mm2,1,2))
          r1=abs(x(mm1,1,3)-x(mm2,1,3))
          if(p1.gt.0.5)p1=p1-1.
          if(q1.gt.0.5)q1=q1-1.
          if(r1.gt.0.5)r1=r1-1.
          rr=met(1,1)*p1*p1+met(2,2)*q1*q1+met(3,3)*r1*r1
          1+met(1,2)*p1*q1+met(1,3)*p1*r1+met(2,3)*q1*r1
          if(rr.lt.f(1,mcop(1))) go to 4502
300    continue
302    continue

```

- Random positioning of the next cations (random choice of M or M') in respect of the distance restraints with the previous ones, on a general or special position, itself selected randomly.

```

C Select randomly a M atom for adding its next neighbour
3500  m=int(randi(iseed)*float(itot)+1.)
      memo=m
C Decide first which positions would generate too much M atoms
C      and eliminate them
      call toomuch(itot,npos,nas,v,fdmax3,np2,np1,pp1)
      if(np2.eq.0)go to 5002
C Decide for the polyhedra type
      mcop(itot+1)=int(randi(iseed)*float(npol)+1.)
      if(mcop(itot+1).gt.npol)mcop(itot+1)=npol
C Determine how many neighbours ? And according to that,
C select the appropriate treatment :
C 1- if ncop(1)=4 then
C   if mv(m,1)=4 : atom already completed

```

```

C   if mv(m,1)=3, 2, or 1 : one atom to add in correct position
C 2- if ncop(1)=6 then
C   if mv(m,1)=6 : atom already completed
C   if mv(m,1)=5, 4, 3, 2, or 1 : one atom to add in correct position
      if(ncop(mcop(m)).eq.3)go to (3001,3002,3500)mv(m,1)
      if(ncop(mcop(m)).eq.4)go to (3001,3002,3003,3500)mv(m,1)
      if(ncop(mcop(m)).eq.5)go to (3001,3002,3003,3004,3500)mv(m,1)
      if(ncop(mcop(m)).eq.6)go to (3001,3002,3003,3004,3005,3500)mv(m,1)
C Case with only one previous neighbour
3001 continue
      gen22=0.
C Decide for the Wyckoff position selected between np1 and npos
600  mwyc(itot+1)=int(randi(iseed)*float(npos-np1)+pp1)
      p=(x(m,1,1)-xa)+xa2*randi(iseed)
      q=(x(m,1,2)-xb)+xb2*randi(iseed)
      r=(x(m,1,3)-xc)+xc2*randi(iseed)
      gen=gen+1.
      gen22=gen22+1.
      if(gen.gt.genmax)go to 5002
C Extend to all positions corresponding to mwyc(nl(1))
..... Etc.
C Case with already 2 previous neighbours
3002 continue
..... Etc
C Case with already 3 neighbours
3003 continue
..... Etc

```

- If a model fulfills all distance restraints, place the X atoms at M-M midpoints, refine the atomic positions and cell parameters so as to improve an R factor (called *R_{dt}* below).

```

C   Place the X atoms and then refine by Monte Carlo
      imemnl=nl(1)
      nl(1)=itot
      call midpt(x,nv,mv,nl,ncop,mcop)
.....Etc.
C Monte Carlo distance improvement
C   Loop of idls moves per atom
C       but move also the cell parameters
      improve=0
      imove=0
      nltot=nl(1)+nl(2)
      mc=idls*nltot
      do 7500 imc=1,mc
C Select an atom
C       i for type 1 or 2
C       m for atom order in the list of either nl(1) or nl(2)
C or select one cell parameter (icel=1)
      icel=int(randi(iseed)*2.)
C Do not refine the cell if iref = 0
      if(iref.eq.0)icel=0
C change a cell parameter a or b or c by (+ or -) 0.01 A max
C change an angle cell parameter alp, bet or gam
C       by (+ or -) 0.01 ° max
      if(icel.eq.1)then
        if(nsym.le.4)mc=int(randi(iseed)*3.+1.)
        if(nsym.eq.5)mc=int(randi(iseed)*4.+1.)
        if(nsym.eq.6)mc=int(randi(iseed)*6.+1.)
C Redefine the cell parameters
      if(mc.eq.1)anew=a+(randi(iseed)-0.5)*0.02
      if(mc.eq.2)bnew=b+(randi(iseed)-0.5)*0.02
      if(mc.eq.3)cnew=c+(randi(iseed)-0.5)*0.02
      if(mc.eq.4)betnew=bet+(randi(iseed)-0.5)*0.02
      if(mc.eq.5)alpnew=alp+(randi(iseed)-0.5)*0.02
      if(mc.eq.6)gamnew=gam+(randi(iseed)-0.5)*0.02
      if(nsym.eq.1)then
        if(mc.eq.1)then
          bnew=anew
          cnew=anew
        endif
      if(mc.eq.2)then

```

```

    anew=bnew
    cnew=bnew
endif
if(mc.eq.3)then
    anew=cnew
    bnew=cnew
endif
alpnew=alp
betnew=bet
gamnew=gam
go to 8001
endif
if(nsym.eq.2.or.nsym.eq.3)then
.....Etc
C Orthorhombic or more
    if(nsym.ge.4)then
C If RdT improved, keep the move...
    rdtnew=sqrt((rd1new+rd2new+rd3new)/
    1(rd1dnew+rd2dnew+rd3dnew))
    if(rdtnew.ge.rdt)go to 7500
    improve=improve+1
C Move kept, then make all changes...
C changes accepted either on cell or atom moves
    if(icel.eq.1)then
C Here modif on cell if accepted
        a=anew
        b=bnew
        c=cnew
        alp=alpnew
        bet=betnew
        gam=gamnew
    else
C changes on coordinates x and Rd1,Rd2 and Rdt
        x(m,i,1)=pmc
        x(m,i,2)=qmc
        x(m,i,3)=rmc
C changes on y
        if(i.eq.1)then
            do ki=1,27*n1(1)
            if(ny(ki).eq.m)then
                y(ki,1)=y(ki,1)+dp
                y(ki,2)=y(ki,2)+dq
                y(ki,3)=y(ki,3)+dr
            endif
            enddo
        endif
C Also changes on xy if i=2...
        if(i.eq.2)then
            do ki=1,n1(1)
            do kj=1,ncop(mcop(ki))
            if(nxy(ki,kj).eq.m)then
                xy(ki,1,kj)=xy(ki,1,kj)+dp
                xy(ki,2,kj)=xy(ki,2,kj)+dq
                xy(ki,3,kj)=xy(ki,3,kj)+dr
            endif
            enddo
        endif
C changes on coordinates x and Rd1,Rd2,Rd3 and Rdt
        rd1=rd1new
        rd2=rd2new
        rd3=rd3new
        rdt=rdtnew
    endif
C write(10,*)'i,m ',i,m
7500 continue
C End of Monte Carlo distance improvement
C If Rdt > Rdtmax, reject the cell
    if(rdt.gt.rdtm)then
        igood=igood-1
        go to 5001
    endif

```

```

C   Do not save if the framework density is outside
C       of the expected range
      if(v.lt.1.)then
        igood=igood-1
        go to 5001
      endif
      tn=nl(1)
      rho(igood)=tn/v*1000.
      if(rho(igood).lt.fDMIN)then
        igood=igood-1
        go to 5001
      endif
      if(rho(igood).gt.fDmax)then
        igood=igood-1
        go to 5001
      endif

```

- Continue to try to predict structures in that way till a certain number of independent runs are made. Verify if the predicted structures are new or were already described (using CS - Coordination Sequence, a fingerprint of the structure).

```

C   Calculate coordination sequence
      call coorseq(nl,x,met,g,ntype,nr,ns,jmax,ncop,mcop,npol)
.....Etc
C   Try to identify if this is already known or already predicted
C   Now compare with data in connectivity.txt ...
C   prepare the current data
      iprint=1
      do i=1,ntype
        do j=1,10
          write(t(j),'(i4)')nr(i,j)
        enddo
        newcos(i,igood)=t(1)//t(2)//t(3)//t(4)//t(5)//t(6)//t(7)//
1t(8)//t(9)//t(10)
      enddo
C   compare only on the real length of the predicted sequence
      l=4*jmax
      do 6800 i=1,ndat
C   skip if not same ntype
        if(nzeot(i).ne.ntype)go to 6800
        isum=0
        do 6700 j=1,ntype
          do 6699 k=1,nzeot(i)
            if(newcos(j,igood)(1:l).eq.coseq(k,i)(1:l))then
              isum=isum+1
            if(isum.eq.ntype)go to 6801
            endif
6699      continue
6700      continue
6800      continue
C   Nothing found
        go to 6810
C   Something found
6801      continue
        ident(igood)=zeot(i)
        write(10,*)
        write(10,*)'This is probably ',zeot(i)
        iprint=0
C   but make the output files .cif, .dat and .xtl anyway
C       if this is asked for (isave=1)
          if(isave.eq.1)iprint=1
          go to 6820
6810      continue
C   Compare also with current list
      do 6850 i=1,igood-1
C   skip if not same ntype
        if(ntype.ne.mntype(i))go to 6850
        isum=0
        jmaxmin=jmax
        if(mjmax(i).lt.jmaxmin)jmaxmin=mjmax(i)
        l=4*jmaxmin

```

```

do 6750 j=1,ntype
do 6749 k=1,mntype(i)
if(newcos(j,igood)(1:1).eq.newcos(k,i)(1:1))then
isum=isum+1
  if(isum.eq.ntype)then
    if(rd(igood).gt.rdp(i))iprint=0
    if(rd(igood).lt.rdp(i))rdp(i)=rd(igood)
    go to 6851
  endif
go to 6750
endif
6749 continue
6750 continue
6850 continue
C   Nothing found
    go to 6860
C   Something found
6851 continue
    ident(igood)=ident(i)
    write(10,*)
    write(10,*)'This was found already ',ident(i)
    go to 6820
6860 continue
C   If nothing found, this is a new one...
    inew=inew+1
    write(ident(igood),'(a4,i7)')'PCOD',inew
    write(10,*)
    write(10,*)'Found for the first time ',ident(igood)
6820 continue
C   End of checking

```

In the GRINSP algorithm, the number of M or M' atoms in a randomly selected cell is not predetermined, it is predicted as well. Only restraints on distances are considered (not angles - though considering a range for the second M-M distances is like restraining angles).


Currently, there are some limitations in that version. GRINSP 1.00 proved to be relatively efficient for a maximum number of 64 M/M' atoms on up to 1-10 different general or special positions. It was possible to retrieve many known zeolites (ABW, ACO, AFI, ANA, AST, BIK, CAN, EAB, EDI, GIS, GME, LOS, LTA, MEP, SOD...) and the compact SiO₂ phases (quartz, cristobalite, tridymite, etc), polymorphs for B₂O₃, AlF₃, hypothetical phases in binary systems B₂O₃/SiO₂, B₂O₃/ReO₃, SiO₂/ReO₃ (see the PCOD [9], a database accumulating these predicted structures). It is up to you to try GRINSP with other systems, and even the above ones have not been completely explored (in part due to that limitation to 64 M/M' atoms and because the maximum cell parameters were generally set to 16 Å). One life would not be sufficient if one formulation explored for one space group needs one or several days of calculations on a standard PC.

Further work is needed for improving the GRINSP efficiency : introduction of different linkage modes than by corners (edges, faces...) but this would mean that all X atoms could not be placed at the M-M mid points; adding the possibility for insertion of big cations K/Sr/Ba/Cs/etc as spheres in the holes/tunnels; considering bond valence as an alternative to pure geometrical restraints for the model final refinements; increase the speed by not recalculating always everything (distances); increase the box size for the CS (coordination sequence) calculations (the 729 cells used are not always enough); increase that 64 M/M' atoms limit; allow to select the space group randomly as well; optimize the code; etc !


These improvements would need faster microprocessors or using a grid of computers on the internet. Anyway, the main problem is that once a model is built, it has to be checked visually. The process is not yet completely automatized (my confidence in it is not absolute, some two-dimensional models have to be discarded, etc).

A few results with 3D 6-connected frameworks


All the known varieties of AlF_3 (pyrochlore, perovskite, HTB...) are predicted by GRINSP, including this strange $\tau\text{-AlF}_3$. New varieties are to be expected, if the GRINSP predictions are confirmed. Mixed compounds with two octahedra sizes were also modelled. Some are known, other are not. In some cases, the chemical composition is enough precise for suggesting the synthesis (contrarily to a simple MX_3 or MX_2 formula), see the figures below. Work is in progress for trying to confirm some of these predictions.




PCOD1000015
 $[\text{Ca}_4\text{Al}_7\text{F}_{33}]^{4-}$, cubic, $a = 10.860 \text{ \AA}$.
 Known with Na atoms in the holes,
 as $\text{Na}_4\text{Ca}_4\text{Al}_7\text{F}_{33}$



PCOD1000017
 AlF_3 , cubic, $a = 9.668 \text{ \AA}$.
 Known with some OH^- and water
 in the holes : pyrochlore



PCOD1010005
 $[\text{Ca}_3\text{Al}_4\text{F}_{21}]^{3-}$, cubic, $a = 9.009 \text{ \AA}$.
UNKNOWN
 Could be stuffed by Na atoms and give the
 hypothetical $\text{Na}_3\text{Ca}_3\text{Al}_4\text{F}_{21}$ - or stuffed by Li atoms



PCOD1000014
 AlF_3 , tetragonal,
 $a = 10.216 \text{ \AA}$, $c = 7.241 \text{ \AA}$.
 Known as $\tau\text{-AlF}_3$

The above pictures are screen copies from drawings made by using a VRML visualizer (CosmoPlayer) reading the .wrl files stored in the PCOD. These .wrl files were made by the STRUVIR program from the .dat files also available in PCOD (as well as CIF files). GRINSP itself has no graphical user interface (GUI), which may not be absolutely necessary if one considers the simplicity of the data necessary for a prediction :


Test P-62c - 190	! Title
P -6 2 C	! Space group
3 1 0 2	! Codes for cell symm., 1 or 2 M atom types, icon, min nber of M
4	! Coordination of the M atom(s)


```

Si  O      ! Definition of the MX couple(s)
6. 16. 6. 16. 6. 16. ! Min and max a, b, c
90. 90. 90. 90. 120. 120. ! Min and max angles
5. 30.      ! Min and max framework density
200000 300000 0.02 ! Monte Carlo runs, Monte Carlo events per run, Rdtmax
10000 1      ! Monte Carlo events for x,y,z refinement, cell refined or not
1900000      ! First filename

```

Concerning the sixfold coordination, GRINSP can produce it randomly as octahedra, trigonal prisms or pyramids with a pentagonal base. Moreover, if the tolerance factor R (Rdt in the code) is above 1% , then these polyhedra can be more or less distorted. Some predictions are showing octahedra/trigonal prisms or octahedra/pyramids mixtures. Fancy predictions with large tunnels or huge cavities are sometimes proposed, such as these two examples (on the left, octahedra and trigonal prisms, on the right, octahedra and pyramids mixtures, the pentagonal base of the pyramids covering the tunnel walls) :



Conclusion

Structure prediction is certainly a promising approach, and an unavoidable part of our future in crystallography. It would have to be combined with an efficient prediction of the physical properties and, more difficult, with the prediction of a synthesis procedure... Storing and managing the huge quantity of hypothetical phases will be a problem, and the one structure/one publication scheme applied for the real compounds will not be relevant to predicted compounds. With 540 predicted structures from GRINSP, PCOD is a dwarf compared to a brand new database of hypothetical zeolites [10] containing in 3 parts (at the time of writing this paper) : 114010 structures in the Bronze database (raw predictions), 33652 refined structures in the Silver database, but nothing yet in the Gold database which will contain unique models. PCOD contains already unique models (almost), and is not restricted to the SiO_2 formulation. The number of SiO_2 predicted polymorphs in the PCOD continues to increase by the slow exploration of all space groups with GRINSP (only the triclinic and cubic space groups were examined systematically yet). The final number of structures depends on the limit fixed on R for retaining or not a model. The R value was arbitrarily chosen to be smaller than 1% in the SiO_2 case. This allows to produce the observed zeolites and the known dense SiO_2 phases, and will finally add a few thousands of hypothetical ones when the exploration will be completed. The MX_3 3D 6-connected hypothetical frameworks will certainly be much less numerous. The $\text{SiO}_2/\text{B}_2\text{O}_3$ system was found even richer than for the simple SiO_2 formulation, even limiting R below 0.6%, though there is not any $\text{B}_x\text{Si}_y\text{O}_z$ in the ICSD database which would include BO_3 triangles and SiO_4 tetrahedra interconnected by corners ! So, what to do with all these predictions now ?

References

- [1] A. Le Bail, J.L. Fourquet and U. Bentrup, *J. Solid State Chem.* **100** (1992) 151-159.
- [2] O. Delgado Friedrichs, A.W.M. Dress, D.H. Huson,, J. Klinowski, A.L. Mackay, *Nature* **400** (1999) 644.
- [3] M.D. Foster, O. Delgado Friedrichs, R.G. Bell, F.A. Almeida Paz, J. Klinowski, *Angew Chem. Int. Ed.* **42** (2003) 3896-3899.
- [4] Systematic enumeration of crystalline networks : <http://www-klinowski.ch.cam.ac.uk/SistEnumHome.htm>

- [5] J.V. Smith, Chem. Rev. 88 (1988) 149-182.
- [6] M.W. Deem and J.M. Newsam, J. Am. Chem. Soc. 114 (1992) 7189-7198.
- [7] A.F. Wells (1977). Three-dimensional Nets and Polyhedra (Wiley-Interscience, New York).
- [8] GRINSP : <http://sdpd.univ-lemans.fr/grinsp/>
- [9] PCOD - Predicted Crystallography Open Database: <http://www.crystallography.net/pcod/>
- [10] Database of Hypothetical Zeolites : <http://www.hypotheticalzeolites.net/>